NAG Fortran Library Routine Document

F08SCF (DSYGVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08SCF (DSYGVD) computes all the eigenvalues, and optionally, the eigenvectors of a real generalized symmetric-definite eigenproblem, of the form

$$Az = \lambda Bz, \quad ABz = \lambda z \quad \text{or} \quad BAz = \lambda z,$$

where A and B are symmetric and B is also positive-definite. If eigenvectors are desired, it uses a divide-and-conquer algorithm.

2 Specification

```
      SUBROUTINE F08SCF (ITYPE, JOBZ, UPLO, N, A, LDA, B, LDB, W, WORK, LWORK,
     1                   IWORK, LIWORK, INFO)
      INTEGER            ITYPE, N, LDA, LDB, LWORK, IWORK(*), LIWORK, INFO
      double precision   A(LDA,*), B(LDB,*), W(*), WORK(*)
      CHARACTER*1        JOBZ, UPLO
```

The routine may be called by its LAPACK name ***dsygvd***.

3 Description

F08SCF (DSYGVD) first performs a Cholesky factorization of the matrix B as $B = U^TU$, when UPLO = 'U' or $B = LL^T$, when UPLO = 'L'. The generalized problem is then reduced to a standard symmetric eigenvalue problem

$$Cx = \lambda x,$$

which is solved for the eigenvalues and, optionally, the eigenvectors; the eigenvectors are then backtransformed to give the eigenvectors of the original problem.

For the problem $Az = \lambda Bz$, the eigenvectors are normalized so that the matrix of eigenvectors, z, satisfies

$$Z^TAZ = \Lambda \quad \text{and} \quad Z^TBZ = I,$$

where Λ is the diagonal matrix whose diagonal elements are the eigenvalues. For the problem $ABz = \lambda z$ we correspondingly have

$$Z^{-1}AZ^{-T} = \Lambda \quad \text{and} \quad Z^TBZ = I,$$

and for $BAz = \lambda z$ we have

$$Z^TAZ = \Lambda \quad \text{and} \quad Z^TB^{-1}Z = I.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: ITYPE – INTEGER *Input*

On entry: specifies the problem type to be solved:

if ITYPE = 1, $Az = \lambda Bz$;
if ITYPE = 2, $ABz = \lambda z$;
if ITYPE = 3, $BAz = \lambda z$.

2: JOBZ – CHARACTER*1 *Input*

On entry: if JOBZ = 'N', compute eigenvalues only.

If JOBZ = 'V', compute eigenvalues and eigenvectors.

Constraint: JOBZ = 'N' or 'V'.

3: UPLO – CHARACTER*1 *Input*

On entry: if UPLO = 'U', the upper triangles of A and B are stored.

If UPLO = 'L', the lower triangles of A and B are stored.

4: N – INTEGER *Input*

On entry: n, the order of the matrices A and B.

Constraint: N $\geq$ 0.

5: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the symmetric matrix A.

If UPLO = 'U', the leading n by n upper triangular part of A contains the upper triangular part of the matrix A.

If UPLO = 'L', the leading n by n lower triangular part of A contains the lower triangular part of the matrix A.

On exit: if JOBZ = 'V', then if INFO = 0, A contains the matrix Z of eigenvectors. The eigenvectors are normalized as follows:

if ITYPE = 1 or 2, $Z^TBZ = I$;
if ITYPE = 3, $Z^TB^{-1}Z = I$.

If JOBZ = 'N', the upper triangle (if UPLO = 'U') or the lower triangle (if UPLO = 'L') of A, including the diagonal, is destroyed.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08SCF (DSYGVD) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

7: B(LDB,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{N})$.

On entry: the symmetric matrix B:

if UPLO = 'U', the leading n by n upper triangular part of B contains the upper triangular part of the matrix B;
if UPLO = 'L', the leading n by n lower triangular part of B contains the lower triangular part of the matrix B.

On exit: if INFO $\leq$ N, the part of B containing the matrix is overwritten by the triangular factor U or L from the Cholesky factorization $B = U^T U$ or $B = LL^T$.

8: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08SCF (DSYGVD) is called.

Constraint: LDB $\geq \max(1, N)$.

9: W($*$) – ***double precision*** array *Output*

Note: the dimension of the array W must be at least $\max(1, N)$.

On exit: if INFO $= 0$, the eigenvalues in ascending order.

10: WORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) returns the optimal LWORK.

11: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08SCF (DSYGVD) is called.

For good performance, LWORK should usually be larger than the minimum, try increasing by $nb \times N$, where nb is the block size.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if N ≤ 1, LWORK ≥ 1;
if JOBZ = 'N' and N > 1, LWORK $\geq 2 \times N + 1$;
if JOBZ = 'V' and N > 1, LWORK $\geq 1 + 6 \times N + 2 \times N^2$.

12: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{LIWORK})$.

On exit: if INFO $= 0$, IWORK(1) returns the optimal LIWORK.

13: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08SCF (DSYGVD) is called.

If LIWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal sizes of the WORK and IWORK arrays, returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK or LIWORK is issued.

Constraints:

if N ≤ 1, LIWORK ≥ 1;
if JOBZ = 'N' and N > 1, LIWORK ≥ 1;
if JOBZ = 'V' and N > 1, LIWORK $\geq 3 + 5 \times N$.

14: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith argument had an illegal value.

INFO > 0

F07FDF (DPOTRF) or F08FCF (DSYEVD) returned an error code:

$\leq$ N if INFO $= i$, F08FCF (DSYEVD) failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero;

$>$ N if INFO $=$ N $+ i$, for $1 \leq i \leq$ N, then the leading minor of order i of B is not positive-definite. The factorization of B could not be completed and no eigenvalues or eigenvectors were computed.

7 Accuracy

If B is ill-conditioned with respect to inversion, then the error bounds for the computed eigenvalues and vectors may be large, although when the diagonal elements of B differ widely in magnitude the eigenvalues and eigenvectors may be less sensitive than the condition of B would suggest. See Section 4.10 of Anderson *et al.* (1999) for details of the error bounds.

The example program below illustrates the computation of approximate error bounds.

8 Further Comments

The total number of floating point operations is proportional to n^3.

The complex analogue of this routine is F08SQF (ZHEGVD).

9 Example

To find all the eigenvalues and eigenvectors of the generalized symmetric eigenproblem $ABz = \lambda z$, where

$$A = \begin{pmatrix} 0.24 & 0.39 & 0.42 & -0.16 \\ 0.39 & -0.11 & 0.79 & 0.63 \\ 0.42 & 0.79 & -0.25 & 0.48 \\ -0.16 & 0.63 & 0.48 & -0.03 \end{pmatrix}$$

and

$$B = \begin{pmatrix} 4.16 & -3.12 & 0.56 & -0.10 \\ -3.12 & 5.03 & -0.83 & 1.09 \\ 0.56 & -0.83 & 0.76 & 0.34 \\ -0.10 & 1.09 & 0.34 & 1.18 \end{pmatrix},$$

together with an estimate of the condition number of B, and approximate error bounds for the computed eigenvalues and eigenvectors.

The example program for F08SAF (DSYGV) illustrates solving a generalized symmetric eigenproblem of the form $Az = \lambda Bz$.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08SCF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDB, LIWORK, LWORK
      PARAMETER        (LDA=NMAX,LDB=NMAX,LIWORK=3+5*NMAX,
     +                 LWORK=1+(NB+6+2*NMAX)*NMAX)
      DOUBLE PRECISION ONE
      PARAMETER        (ONE=1.0D+0)
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM, BNORM, EPS, RCOND, RCONDB, T1, T2, T3
      INTEGER          I, IFAIL, INFO, J, LIWOPT, LWOPT, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(LDB,NMAX), EERBND(NMAX),
     +                 RCONDZ(NMAX), W(NMAX), WORK(LWORK), ZERBND(NMAX)
      INTEGER          IWORK(LIWORK)
*     .. External Functions ..
      DOUBLE PRECISION F06RCF, X02AJF
      EXTERNAL         F06RCF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         DDISNA, DSYGVD, DTRCON, X04CAF
*     .. Intrinsic Functions ..
      INTRINSIC        ABS
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08SCF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the upper triangular parts of the matrices A and B
*
         READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
         READ (NIN,*) ((B(I,J),J=I,N),I=1,N)
*
*        Compute the one-norms of the symmetric matrices A and B
*
         ANORM = F06RCF('One norm','Upper',N,A,LDA,WORK)
         BNORM = F06RCF('One norm','Upper',N,B,LDB,WORK)
*
*        Solve the generalized symmetric eigenvalue problem
*        A*B*x = lambda*x (ITYPE = 2)
*
         CALL DSYGVD(2,'Vectors','Upper',N,A,LDA,B,LDB,W,WORK,LWORK,
     +               IWORK,LIWORK,INFO)
         LWOPT = WORK(1)
         LIWOPT = IWORK(1)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Eigenvalues'
            WRITE (NOUT,99999) (W(J),J=1,N)
*
            IFAIL = 0
            CALL X04CAF('General',' ',N,N,A,LDA,'Eigenvectors',IFAIL)
*
*           Call DTRCON (F07TGF) to estimate the reciprocal condition
*           number of the Cholesky factor of B.  Note that:
*           cond(B) = 1/RCOND**2
```

```
*
            CALL DTRCON('One norm','Upper','Non-unit',N,B,LDB,RCOND,
     +                  WORK,IWORK,INFO)
*
*           Print the reciprocal condition number of B
*
            RCONDB = RCOND**2
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Estimate of reciprocal condition number for B'
            WRITE (NOUT,99998) RCONDB
*
*           Get the machine precision, EPS, and if RCONDB is not less
*           than EPS**2, compute error estimates for the eigenvalues and
*           eigenvectors
*
            EPS = X02AJF()
            IF (RCOND.GE.EPS) THEN
*
*              Call DDISNA (F08FLF) to estimate reciprocal condition
*              numbers for the eigenvectors of (A*B - lambda*I)
*
               CALL DDISNA('Eigenvectors',N,N,W,RCONDZ,INFO)
*
*              Compute the error estimates for the eigenvalues and
*              eigenvectors
*
               T1 = ONE/RCOND
               T2 = EPS*T1
               T3 = ANORM*BNORM
               DO 20 I = 1, N
                  EERBND(I) = EPS*(T3+ABS(W(I))/RCONDB)
                  ZERBND(I) = T2*(T3/RCONDZ(I)+T1)
   20          CONTINUE
*
*              Print the approximate error bounds for the eigenvalues
*              and vectors
*
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Error estimates for the eigenvalues'
               WRITE (NOUT,99998) (EERBND(I),I=1,N)
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'Error estimates for the eigenvectors'
               WRITE (NOUT,99998) (ZERBND(I),I=1,N)
            ELSE
               WRITE (NOUT,*)
               WRITE (NOUT,*) 'B is very ill-conditioned, error ',
     +           'estimates have not been computed'
            END IF
         ELSE IF (INFO.GT.N .AND. INFO.LE.2*N) THEN
            I = INFO - N
            WRITE (NOUT,99997) 'The leading minor of order ', I,
     +        ' of B is not positive definite'
         ELSE
            WRITE (NOUT,99996) 'Failure in DSYGVD. INFO =', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99995) 'Optimum workspace required = ', LWOPT,
     +        'Workspace provided         = ', LWORK
         END IF
         IF (LIWORK.LT.LIWOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99995) 'Integer workspace required = ', LIWOPT,
     +        'Integer workspace provided = ', LIWORK
         END IF
      ELSE
         WRITE (NOUT,*) 'NMAX too small'
```

```
      END IF
      STOP
*
99999 FORMAT (3X,(6F11.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4,A)
99996 FORMAT (1X,A,I4)
99995 FORMAT (1X,A,I5,/1X,A,I5)
      END
```

9.2 Program Data

```
F08SCF Example Program Data

  4                         :Value of N

  0.24   0.39   0.42  -0.16
        -0.11   0.79   0.63
               -0.25   0.48
                      -0.03 :End of matrix A

  4.16  -3.12   0.56  -0.10
         5.03  -0.83   1.09
                0.76   0.34
                       1.18 :End of matrix B
```

9.3 Program Results

```
 F08SCF Example Program Results

 Eigenvalues
      -3.5411    -0.3347     0.2983     2.2544
 Eigenvectors
            1          2          3          4
 1      0.0356     0.1039    -0.7459     0.1909
 2     -0.3809    -0.4322    -0.7845     0.3540
 3      0.2943    -1.5644    -0.7144     0.5665
 4      0.3186     1.0647     1.1184     0.3859

 Estimate of reciprocal condition number for B
        5.8E-03

 Error estimates for the eigenvalues
        7.0E-14    8.6E-15    7.9E-15    4.6E-14

 Error estimates for the eigenvectors
        2.8E-14    6.4E-14    6.4E-14    3.4E-14
```